

BATCH LEAST SQUARES DIFFERENTIAL CORRECTION  
OF A HELIOCENTRIC ORBIT

PART 1 - TEST CASE SPECIFICATION WORKSHEET

Roger L. Mansfield, June 21, 2016  
<http://astroger.com>

This worksheet defines a test case for the HDC worksheet. HDC implements the equations of weighted batch least squares differential correction (DC) of a heliocentric orbit.

This worksheet inputs observations of the orbit of a minor planet or asteroid from the file specified below, which replaces the generic file OBSERVATIONS.PRN.

Here now is an outline of the steps we will follow in this worksheet:

1. Specify the observations and the weight matrix, W.
2. Specify the observer's coordinates and calculate the observer's ECI equatorial positions at the observation times.
3. Calculate the sun's ECI equatorial coordinates and then the sun's coordinates relative to the observer at the observation times.
4. Calculate the N-by-1 "observed" measurements vector, Y.
5. Specify the initial estimate of state,  $X_0$ .
6. Write test case specification values to disk for use by worksheet HDC.

(To create other test cases for HDC, simply duplicate this worksheet, rename as "HD2", "HD3", etc., and modify as desired. HDC works best with test cases specified in this way.)

As a preliminary, we define some constants that we will need and set the Mathcad worksheet ORIGIN to 1 so that subscripts start at unity rather than at zero.

$$\text{DegPerRad} := \frac{180}{\pi}$$

$$\text{ORIGIN} \equiv 1$$

$$\text{SecPerDeg} := 3600.0$$

$$\text{SecPerRad} := \text{SecPerDeg} \cdot \frac{180}{\pi}$$

$$\text{SecPerRev} := \text{SecPerDeg} \cdot 360.0$$

TIPS ON READING MATHCAD WORKSHEETS

1. A Mathcad worksheet typically consists of text regions and math regions.
2. Text and math regions can be anywhere on a page. Text regions are just optional comments. Mathcad uses math regions to do its calculations, and according to the following rule: math regions are calculated in the order of left to right, then top to bottom.
3. In a math region, colon-equals ( := ) is like an assignment statement in C. That is, the expression on the right is calculated and placed in the variable on the left, whereas equals ( = ) by itself is used to display on the right of the equals sign the value of the variable on the left.
4. Mathcad has functions just like C does. Typically a function has its name and input variables inside of parentheses on the left of a colon-equals ( := ) and a sequence of vertical line segments with the assignment statements of the function on the right. Inside of a function, the assignment statements use left arrows <-- instead of colon-equals := to make assignments. The last line of a function is its output argument, which may be a scalar, a vector, or a matrix of variables.
5. My Mathcad worksheets generally are formatted so that the flow is left to right, top to bottom of an 8.5" x 11" page when the worksheet is printed.
  - But I like to use a second 8.5" page side-by-side for additional material, added later, that I might later delete, e.g., scratchpad or temporary calculations.
  - Sometimes I use this extra right-margin space to do, in parallel, calculations needed later on, or to simply add material without disturbing the main flow.
  - When my worksheet has the second, side-by-side page and I want to publish the worksheet, I print the worksheet as an Adobe .pdf file. I specify the "ledger" format, which, thankfully, prints out both pages side-by-side.

These tips themselves fit my second, side-by-side page convention: they are additionaly material that is not part of the main flow of the worksheet.

1. Input the observations and sensor coordinates and specify the weight matrix, W.

Specify time (**JDT**), right ascension (**RA**), and declination (**DEC**) for observations 1 through n. Note that **RA** and **DEC** are referred to the true equator and equinox of 1801 January 1.

```
Obs := READPRN("1801_Ceres_with_LMT+12h_minus_obs_3,6.txt")
```

```

1801 1 1 20 43 17.8 51 47 48.8 15 37 43.5 535
1801 1 2 20 39 4.6 51 43 27.8 15 41 5.5 535
1801 1 4 20 30 42.1 51 35 47.3 15 47 57.6 535
1801 1 10 20 6 15.8 51 23 1.5 16 10 32 535
1801 1 14 19 50 31.7 51 22 55.8 16 27 5.7 535
1801 1 19 19 31 28.5 51 32 2.3 16 49 16.1 535
1801 1 21 19 24 2.7 51 38 34.1 16 58 35.9 535
1801 1 22 19 20 21.7 51 42 21.3 17 3 18.5 535
Obs =
1801 1 23 19 16 43.5 51 46 43.5 17 8 5.5 535
1801 1 28 18 58 51.3 52 13 38.3 17 32 54.1 535
1801 1 30 18 51 52.9 52 27 2.1 17 43 11 535
1801 1 31 18 48 26.4 52 34 18.8 17 48 21.5 535
1801 2 1 18 44 59.9 52 41 48 17 53 36.5 535
1801 2 2 18 41 35.8 52 49 45.9 17 58 57.5 535
1801 2 5 18 31 31.5 53 15 40.5 18 15 1 535
1801 2 8 18 21 39.2 53 44 37.5 18 31 23.2 535
1801 2 11 18 11 58.2 54 16 38.1 18 47 58.8 535

```

Retrieve observations matrix from text file.

Then set observation n and measurement count N:

```
n := rows(Obs)
```

```
N := 2·n
```

We define and then invoke the procedural functions **JED19**, **AzRA**, and **EIDec** to extract and convert the **JDT**, **AZRA**, and **ELDEC** observation vectors, respectively, for n observations.

**DayCount** specifies the count of days from the beginning of the year, up through the last day of the previous month of any non-leap year. **JED18** calculates the number of Julian ephemeris days corresponding to Year, Month, and Day. Note that **JED18** is intended to be used with any Gregorian calendar date since 1800 January 0.0, having JED = 2378495.5.

**DayCount** := (0 31 59 90 120 151 181 212 243 273 304 334)<sup>T</sup>

**JED18**(Year, Month, Day) :=

```

JED ← 2378495.5
for Y ∈ 1800 .. Year
    if Y < Year
        JED ← JED + 365 if mod(Y, 4) ≠ 0
    otherwise
        if mod(Y, 100) = 0
            JED ← JED + 366 if mod(Y, 400) = 0
            JED ← JED + 365 otherwise
        JED ← JED + 366 otherwise
    otherwise
        JED ← JED + DayCountMonth + Day
        JED ← JED + 0 if mod(Y, 4) ≠ 0
    otherwise
        if Month > 2
            if mod(Y, 100) = 0
                JED ← JED + 1 if mod(Y, 400) = 0
                JED ← JED + 0 otherwise
            JED ← JED + 1 otherwise
        JED ← JED + 0 otherwise
    JED

```

**JDTCalc**(k) :=

```

for i ∈ 1 .. k
    JDTi ← JED18(Obsi, 1, Obsi, 2, Obsi, 3)
    JDTi ← JDTi + Obsi, 4 / 24 + Obsi, 5 / 1440 + Obsi, 6 / 86400
JDT

```

**JDT** := **JDTCalc**(n)

Epoch := **JDT**<sub>1</sub>

Epoch = 2378862.36340046

Assume in this worksheet that RA is expressed  
in deg/min/sec rather than hours/min/sec:

$$\text{AzRA}(k) := \begin{cases} \text{for } i \in 1..k \\ \quad \text{AZRA}_i \leftarrow \frac{\text{Obs}_{i,7} + \frac{\text{Obs}_{i,8}}{60} + \frac{\text{Obs}_{i,9}}{3600}}{\text{DegPerRad}} \\ \text{AZRA} \end{cases}$$

**RA** := AzRA(n)      Right ascensions are in radians.

$$\text{ElDec}(k) := \begin{cases} \text{for } i \in 1..k \\ \quad \text{ELDEC}_i \leftarrow \frac{\text{Obs}_{i,10} + \frac{\text{Obs}_{i,11}}{60} + \frac{\text{Obs}_{i,12}}{3600}}{\text{DegPerRad}} \\ \quad \text{ELDEC}_i \leftarrow -\text{ELDEC}_i \text{ if } \text{Obs}_{i,10} < 0 \\ \text{ELDEC} \end{cases}$$

**DEC** := ElDec(n)      Declinations are in radians.

$$\text{WEIGHT}(k) := \begin{cases} \text{for } i \in 1..k \\ \quad \text{for } j \in 1..k \\ \quad \quad W_{i,j} \leftarrow 1 \text{ if } i = j \\ \quad \quad W_{i,j} \leftarrow 0 \text{ otherwise} \\ W \end{cases}$$

**W** := WEIGHT(N)

2. Input the observer's geographical coordinates\* in MPC format.

**Sensor** := READPRN("Observers.txt")      NSen := rows(**Sensor**)

$$\text{Sensor} = \begin{pmatrix} 283 & 8.8163 & 0.60204 & 0.79579 \\ 528 & 9.9426 & 0.6234 & 0.77931 \\ 280 & 8.9118 & 0.60114 & 0.79646 \\ 535 & 13.3578 & 0.78782 & 0.61386 \end{pmatrix} \quad \text{NSen} = 4$$

Compute the observer's ("sensor's") ECI equatorial positions at the observation times via function **SENPOS**.

We know that Piazzi used a meridian circle to make his observations of Ceres. Since the right ascension of a target celestial body, in hours, as measured by a meridian circle, is the local apparent sidereal time of the observer, we take advantage of this fact when we compute the right ascensions of the observer in **SENPOS**.

```
SENPOS(k) := | for i ∈ 1 .. k
                  |   θ ← RAi
                  |   for j ∈ 1 .. NSen
                  |     if Sensorj, 1 = Obsi, 9
                  |       RC ← Sensorj, 3
                  |       RS ← Sensorj, 4
                  |       λ ← Sensorj, 2 / DegPerRad
                  |       Ri ←  $\begin{pmatrix} RC \cdot \cos(\theta) \\ RC \cdot \sin(\theta) \\ RS \end{pmatrix}$ 
                  |
                  | R
```

Note that  $\lambda$  is computed but not used in **SENPOS**.

This is because a meridian circle measures the right ascension of the celestial object directly, and this is also the local apparent sidereal time. So the arguments of the cos and sin functions in **SENPOS** become  $(\theta)$  instead of  $(\theta + \lambda)$ .

\*MPC observatory codes and the observers in year 1806:

283 = Bremen (Olbers)  
 528 = Goettingen (Harding)  
 280 = Lilienthal (Bessel)  
 535 = Palermo (Piazzi)

Set  $\Delta$  values to zero to initialize for light-time correction.

```
DELTA(k) := | for i ∈ 1 .. k
                  |   Δi ← 0.0
                  |
                  | Δ
```

$\Delta := \text{DELTA}(N)$

3. Calculate sun's ECI positions at the observation times. Here we will use **C**, **U2PM** and **HGEO** as defined in previous worksheets, for convenience, even though UPM is not really needed for the very-nearly-circular orbit of the Earth around the Sun.

We will need function **C** to calculate the first four c-functions for function **U2PM**.

```
C(x) := N ← 0
  while |x| ≥ 0.1
    x ←  $\frac{x}{4}$ 
    N ← N + 1
     $c_3 \leftarrow \frac{\left[1 - \frac{x}{20}\right]\left[1 - \frac{x}{42}\right]\left[1 - \frac{x}{72}\right]\left[1 - \frac{x}{110}\right]\left[1 - \frac{x}{156}\left(1 - \frac{x}{210}\right)\right]}{6}$ 
     $c_2 \leftarrow \frac{\left[1 - \frac{x}{12}\right]\left[1 - \frac{x}{30}\right]\left[1 - \frac{x}{56}\right]\left[1 - \frac{x}{90}\right]\left[1 - \frac{x}{132}\left(1 - \frac{x}{182}\right)\right]}{2}$ 
     $c_1 \leftarrow 1 - c_3 \cdot x$ 
     $c_0 \leftarrow 1 - c_2 \cdot x$ 
  while N > 0
    N ← N - 1
     $c_3 \leftarrow \frac{(c_1 \cdot c_2 + c_3)}{4}$ 
     $c_2 \leftarrow \frac{c_1 \cdot c_1}{2}$ 
     $c_1 \leftarrow c_1 \cdot c_0$ 
     $c_0 \leftarrow 2 \cdot c_0 \cdot c_0 - 1$ 
   $(c_0 \ c_1 \ c_2 \ c_3)^T$ 
```

We will need the uniform, two-body path propagator function, **U2PM**, which will be invoked by function **HGEO**.

$k := 0.01720209895$

$\mu := 1.0$

$K := k \cdot \sqrt{\mu}$

```

U2PM(K,q,e,i, $\Omega$ , $\omega$ , $\Delta t$ ) := 
   $\alpha \leftarrow K^2 \cdot \frac{(1 - e)}{q}$ 
   $p \leftarrow q \cdot (1 + e)$ 
   $s \leftarrow \frac{\Delta t}{q}$ 
   $\Delta s \leftarrow s$ 
  while  $|\Delta s| \geq 0.00000001$ 
     $c \leftarrow C(\alpha \cdot s^2)$ 
     $f \leftarrow q \cdot s + K^2 \cdot e \cdot s^3 \cdot c_4 - \Delta t$ 
     $Df \leftarrow q + K^2 \cdot e \cdot s^2 \cdot c_3$ 
     $DDf \leftarrow K^2 \cdot e \cdot s \cdot c_2$ 
     $m \leftarrow 1$  if  $Df \geq 0$ 
     $m \leftarrow -1$  otherwise
     $\Delta s \leftarrow \frac{-5 \cdot f}{[Df + m \cdot \sqrt{[(4 \cdot Df)^2 - 20 \cdot f \cdot DDf]}]}$ 
     $s \leftarrow s + \Delta s$ 
   $P_1 \leftarrow \cos(\Omega) \cdot \cos(\omega) - \sin(\Omega) \cdot \cos(i) \cdot \sin(\omega)$ 
   $P_2 \leftarrow \sin(\Omega) \cdot \cos(\omega) + \cos(\Omega) \cdot \cos(i) \cdot \sin(\omega)$ 
   $P_3 \leftarrow \sin(i) \cdot \sin(\omega)$ 
   $Q_1 \leftarrow -(\cos(\Omega) \cdot \sin(\omega) + \sin(\Omega) \cdot \cos(i) \cdot \cos(\omega))$ 
   $Q_2 \leftarrow -(\sin(\Omega) \cdot \sin(\omega) - \cos(\Omega) \cdot \cos(i) \cdot \cos(\omega))$ 
   $Q_3 \leftarrow \sin(i) \cdot \cos(\omega)$ 
   $c \leftarrow C(\alpha \cdot s^2)$ 
   $r\cosv \leftarrow q - K^2 \cdot s^2 \cdot c_3$ 
   $r\sinv \leftarrow K \cdot \sqrt{p \cdot s} \cdot c_2$ 
   $r\cosv \cdot P + r\sinv \cdot Q$ 

```

We need function **HGEO** to calculate the ECI equatorial J2000.0 coordinates of the Sun. (**HGEO** actually calculates the heliocentric ecliptic coordinates of the Earth-Moon barycenter, and then corrects them to the geocenter. See [1] for a reference.)

```

HGEO(JD) := | JD0 ← 2451545.0
                | Tc ←  $\frac{JD - JD_0}{36525.0}$ 
                | a ← 1.00000011 - 0.00000005·Tc
                | e ← 0.01671022 - 0.00003804·Tc
                | q ← a·(1 - e)
                | μ ← 1.00000304
                | K ← k· $\sqrt{\mu}$ 
                | n ←  $\frac{K \cdot a^{-3}}{2}$ 
                | ω ←  $\frac{102.94719 + \frac{1198.28 \cdot T_c}{\text{SecPerDeg}}}{\text{DegPerRad}}$ 
                | i ←  $\frac{0.00005 - \frac{46.94 \cdot T_c}{\text{SecPerDeg}}}{\text{DegPerRad}}$ 
                | Ω ← 0.0
                | L ←  $\frac{100.46435 + \frac{1293740.63 + 99 \cdot \text{SecPerRev}}{\text{SecPerDeg}} \cdot T_c}{\text{DegPerRad}}$ 
                | T ← JD -  $\frac{\text{mod}(L - \omega, 2 \cdot \pi)}{n}$ 
                | Δt ← JD - T
                | rEM ← U2PM(K, q, e, i, Ω, ω, Δt)
                | LM ←  $\frac{\text{mod}(218.0 + 481268.0 \cdot T_c, 360.0)}{\text{DegPerRad}}$ 
                | 
$$\begin{pmatrix} r_{EM_1} & = & 0.0000312 \cdot \cos(L_M) \\ r_{EM_2} & = & 0.0000312 \cdot \sin(L_M) \\ r_{EM_3} & = & \end{pmatrix}$$


```

But we will here substitute the improved **HGE01** model. (As noted in Reference 1, Reference 4 provides an improved model. **HGE01** was taken from Reference 4.)

```

HGE01(JD) := | JD0 ← 2451545.0
                  | Tc ←  $\frac{JD - JD_0}{36525.0}$ 
                  | a ← 1.00000261 - 0.00000261·Tc
                  | e ← 0.01671123 - 0.00004392·Tc
                  | q ← a·(1 - e)
                  | μ ← 1.00000304
                  | K ← k· $\sqrt{\mu}$ 
                  | n ←  $\frac{k^3}{K \cdot a^2}$ 
                  | ω ←  $\frac{102.93768193 + 0.32327364 \cdot T_c}{\text{DegPerRad}}$ 
                  | i ←  $\frac{-0.00001531 - 0.01294668 \cdot T_c}{\text{DegPerRad}}$ 
                  | Ω ← 0.0
                  | L ←  $\frac{100.46457166 + 35999.37244981 \cdot T_c}{\text{DegPerRad}}$ 
                  | T ← JD -  $\frac{\text{mod}(L - \omega, 2 \cdot \pi)}{n}$ 
                  | Δt ← JD - T
                  | rEM ← U2PM(K, q, e, i, Ω, ω, Δt)
                  | LM ←  $\frac{\text{mod}(218.0 + 481268.0 \cdot T_c, 360.0)}{\text{DegPerRad}}$ 
                  | 
$$\begin{pmatrix} \mathbf{r}_{EM_1} - 0.0000312 \cdot \cos(L_M) \\ \mathbf{r}_{EM_2} - 0.0000312 \cdot \sin(L_M) \\ \mathbf{r}_{EM_3} \end{pmatrix}$$


```

In order to go from the mean equator and equinox of J2000.0 to the true equator and equinox of 1801 January 1.0, we will need precession and nutation matrices, which will be applied to the Sun's position vector in function **SUNPOS**. **PRECESS1** is the precession model of Capitaine, *et al.* [2].

$$\begin{aligned} \text{PRECESS1}(\mathbf{r}, \text{JD}) := & \left| \begin{array}{l} T \leftarrow \frac{(\text{JD} - 2451545.0)}{36525.0} \\ \psi_A \leftarrow \frac{5038.481507 \cdot T - 1.0790069 \cdot T^2 - 0.00114045 \cdot T^3}{\text{SecPerRad}} \\ \epsilon_o \leftarrow \frac{84381.406}{\text{SecPerRad}} \\ \omega_A \leftarrow \frac{\epsilon_o \cdot \text{SecPerRad} - 0.025754 \cdot T - 0.0512623 \cdot T^2 - 0.00772503 \cdot T^3}{\text{SecPerRad}} \\ \chi_A \leftarrow \frac{10.556403 \cdot T - 2.3814292 \cdot T^2 - 0.00121197 \cdot T^3}{\text{SecPerRad}} \\ S_1 \leftarrow \sin(\epsilon_o) \\ C_1 \leftarrow \cos(\epsilon_o) \\ S_2 \leftarrow \sin(-\psi_A) \\ C_2 \leftarrow \cos(-\psi_A) \\ S_3 \leftarrow \sin(-\omega_A) \\ C_3 \leftarrow \cos(-\omega_A) \\ S_4 \leftarrow \sin(\chi_A) \\ C_4 \leftarrow \cos(\chi_A) \\ \mathbf{P}_{1,1} \leftarrow C_4 \cdot C_2 - S_2 \cdot S_4 \cdot C_3 \\ \mathbf{P}_{1,2} \leftarrow C_4 \cdot S_2 \cdot C_1 + S_4 \cdot C_3 \cdot C_2 \cdot C_1 - S_1 \cdot S_4 \cdot S_3 \\ \mathbf{P}_{1,3} \leftarrow C_4 \cdot S_2 \cdot S_1 + S_4 \cdot C_3 \cdot C_2 \cdot S_1 + C_1 \cdot S_4 \cdot S_3 \\ \mathbf{P}_{2,1} \leftarrow -S_4 \cdot C_2 - S_2 \cdot C_4 \cdot C_3 \\ \mathbf{P}_{2,2} \leftarrow -S_4 \cdot S_2 \cdot C_1 + C_4 \cdot C_3 \cdot C_2 \cdot C_1 - S_1 \cdot C_4 \cdot S_3 \\ \mathbf{P}_{2,3} \leftarrow -S_4 \cdot S_2 \cdot S_1 + C_4 \cdot C_3 \cdot C_2 \cdot S_1 + C_1 \cdot C_4 \cdot S_3 \\ \mathbf{P}_{3,1} \leftarrow S_2 \cdot S_3 \\ \mathbf{P}_{3,2} \leftarrow -S_3 \cdot C_2 \cdot C_1 - S_1 \cdot C_3 \\ \mathbf{P}_{3,3} \leftarrow -S_3 \cdot C_2 \cdot S_1 + C_3 \cdot C_1 \\ \mathbf{P} \cdot \mathbf{r} \end{array} \right| \end{aligned}$$

We define our nutation matrix function, **NUTATE1**, as follows. For a reference, see [3], p. 320.

$$\begin{aligned} \text{NUTATE1}(\mathbf{r}, \text{JD}) := & \left| \begin{array}{l} d \leftarrow \text{JD} - 2451545.0 \\ \varepsilon \leftarrow \frac{23.4392911}{\text{DegPerRad}} \\ \Delta\psi \leftarrow -0.0048 \cdot \sin\left(\frac{125.0 - 0.05295 \cdot d}{\text{DegPerRad}}\right) - 0.0004 \cdot \sin\left(\frac{200.9 + 1.97129 \cdot d}{\text{DegPerRad}}\right) \\ \Delta\psi \leftarrow \frac{\Delta\psi}{\text{DegPerRad}} \\ \Delta\varepsilon \leftarrow 0.0026 \cdot \cos\left(\frac{125.0 - 0.05295 \cdot d}{\text{DegPerRad}}\right) + 0.0002 \cdot \cos\left(\frac{200.9 + 1.97129 \cdot d}{\text{DegPerRad}}\right) \\ \Delta\varepsilon \leftarrow \frac{\Delta\varepsilon}{\text{DegPerRad}} \\ \mathbf{N} \leftarrow \begin{pmatrix} 1.0 & -\Delta\psi \cdot \cos(\varepsilon) & -\Delta\psi \cdot \sin(\varepsilon) \\ \Delta\psi \cdot \cos(\varepsilon) & 1.0 & -\Delta\varepsilon \\ \Delta\psi \cdot \sin(\varepsilon) & \Delta\varepsilon & 1.0 \end{pmatrix} \\ \mathbf{N} \cdot \mathbf{r} \end{array} \right| \end{aligned}$$

We need the obliquity of the ecliptic,  $\varepsilon$ , at J2000.0, in order to transform the ECI ecliptic J2000.0 coordinates of the Sun to ECI equatorial J2000.0 coordinates.

$$\varepsilon := \frac{23.4392911}{\text{DegPerRad}}$$

$$\mathbf{MO} := \begin{pmatrix} 1 & 0 & 0 \\ 0 & \cos(\varepsilon) & -\sin(\varepsilon) \\ 0 & \sin(\varepsilon) & \cos(\varepsilon) \end{pmatrix}$$

$$\mathbf{ECEQ}(\mathbf{r}) := \mathbf{MO} \cdot \mathbf{r}$$

(Transforms from ecliptic to equatorial.)

Define geocentric observer positions **R1** for plot points in HDC worksheet. These are 13 additional equally-space ephemeris points, 30 days apart, starting at **JDT1** epoch.

$$\mathbf{EQEC}(\mathbf{r}) := \mathbf{MO}^{-1} \cdot \mathbf{r}$$

(Transforms from equatorial to ecliptic.)

$$\begin{aligned} \text{SUNPOS}(k) := & \left| \begin{array}{l} \text{for } i \in 1 .. k \\ \mathbf{R}^{(i)} \leftarrow \mathbf{MO} \cdot \text{HGEO1}(\mathbf{JDT}_i) \\ \mathbf{R}^{(i)} \leftarrow \text{NUTATE1}(\text{PRECESS1}(\mathbf{R}^{(i)}, \mathbf{JDT}_i), \mathbf{JDT}_i) \\ -\mathbf{R} \end{array} \right| \end{aligned}$$

$$\begin{aligned} \text{SUNPOS1} := & \left| \begin{array}{l} \text{for } i \in 1 .. 13 \\ t_i \leftarrow \mathbf{JDT}_1 + 30 \cdot (i - 1) \\ \mathbf{R}^{(i)} \leftarrow \mathbf{MO} \cdot \text{HGEO1}(t_i) \\ \mathbf{R}^{(i)} \leftarrow \text{NUTATE1}(\text{PRECESS1}(\mathbf{R}^{(i)}, t_i), t_i) \\ -\mathbf{R} \end{array} \right| \end{aligned}$$

To obtain the ECI equatorial J2000.0 coordinates of the Sun, we compute -1 times the HCl equatorial J2000.0 position vector of Earth (this explains the minus sign of **R** in **SUNPOS**).

**R<sub>SUN</sub>** := **SUNPOS(n)**

	1	2	3	4	5
1	0.18874351	0.20582795	0.23980477	0.33983185	0.40450806
2	-0.88508658	-0.8818886	-0.87467765	-0.84659066	-0.82258615
3	-0.38425418	-0.38286657	-0.37973752	-0.36754842	...

(Contemporary values would agree well with the *Astronomical Almanac*, Section C.)

Calculate the Sun's position vectors relative to the observer at the observation times.

$$\mathbf{R} := \mathbf{R}_{\text{SUN}} - \frac{\mathbf{R}_{\text{SEN}}}{23454.79842}$$

(The constant is the number of Earth radii  
in one A.U.)

4. Calculate the N-by-1 observed measurements vector, Y.

**YVALUES(n)** := for i ∈ 1 .. n

```

j ← 2·i - 1
k ← j + 1
Yj ← cos(DECi)·RAi
Yk ← DECi
Y

```

Y := **YVALUES(n)**

(Note that function **YVALUES** converts n  
observations to N = 2n measurements.)

	1
1	0.87060215
2	0.27277315
3	0.86914497
4	0.27375247
5	0.86650741
6	...

(Click on the column vector Y and  
scroll up or down to see all entries.)

**R1** := **SUNPOS1**

	1	2	3	4	5
1	0.18874351	0.65437884	0.9426055	0.97972169	0.76176569
2	-0.88508658	-0.67612943	-0.28342658	0.18528403	0.6062928
3	-0.38425418	-0.29355964	-0.12308516	0.08040032	...

Write to disk the plot points **R1**:

WRITEPRN("RVALS1.prn") := **R1**

5. Specify the initial estimate of state,  $X_0$ .

We will start with the HCl ecliptic values of position and velocity as obtained from the HGM-Heliocentric worksheets HH1 and HHC.

Note also that because HDC expects position and velocity to be HCl equatorial true of date (TOD), we will need to convert them from ecliptic to equatorial now, and back to ecliptic in worksheet HDC, before we can transform to conic elements there.

$$\mathbf{r} := \begin{pmatrix} 0.96513293 \\ 2.51763452 \\ -0.10511707 \end{pmatrix} \quad \mathbf{v} := \begin{pmatrix} -0.00995586 \\ 0.00301087 \\ 0.00193007 \end{pmatrix}$$

$$\varepsilon := \frac{23.4392911 - 0.0000004 \cdot (\text{JDT}_1 - 2451545.0)}{\text{DegPerRad}} \quad \mathbf{MO} := \begin{pmatrix} 1 & 0 & 0 \\ 0 & \cos(\varepsilon) & -\sin(\varepsilon) \\ 0 & \sin(\varepsilon) & \cos(\varepsilon) \end{pmatrix}$$

$\mathbf{ECEQ}(\mathbf{r}) := \mathbf{MO} \cdot \mathbf{r}$  (Transforms from ecliptic to equatorial.)

$\mathbf{EQEC}(\mathbf{r}) := \mathbf{MO}^{-1} \cdot \mathbf{r}$  (Transforms from equatorial to ecliptic.)

$\mathbf{r} := \mathbf{ECEQ}(\mathbf{r}) \quad \mathbf{v} := \mathbf{ECEQ}(\mathbf{v})$

$X_0 := \text{stack}(\mathbf{r}, \mathbf{v})$

$$X_0 = \begin{pmatrix} 0.96513293 \\ 2.35123815 \\ 0.90620766 \\ -0.00995586 \\ 0.00199317 \\ 0.00296947 \end{pmatrix}$$

(Note that the epoch of this state vector is "Epoch", as defined above, and that the units are A.U. for position and A.U. per day for velocity.)

**6.** Write test case specification values to disk for use by worksheet HDC.

WRITEPRN("NOBS.dat") := n	Number of observations.
WRITEPRN("TVALS.prn") := JDT	Observation times.
WRITEPRN("EPOCH.dat") := Epoch	Epoch of state vector solution.
WRITEPRN("WEIGHTS.prn") := W	Measurement weights matrix.
WRITEPRN("RVALS.prn") := R	Values of R.
WRITEPRN("YVALS.prn") := Y	Values of Y.
WRITEPRN("STATE.prn") := X <sub>0</sub>	State vector (to be corrected by HDC).
WRITEPRN("RMS.prn") := (0 0)	RMS history for state corrections by HDC (one entry for each iteration).
WRITEPRN("DELTA.prn") := Δ	Initialize Δ for light-time corrections. (The array is given dimension N rather than n to make it easy to extract the values from function <b>FXA</b> in worksheet HDC, by augmenting the N-by-1 array that contains <b>FX</b> and A.)

#### REFERENCES

- [1] Seidelmann, P. K. (Editor), *Explanatory Supplement to the Astronomical Almanac*, University Science Books (1992), p. 316. This model was superseded by the model in [4].
- [2] *Astronomical Almanac for the Year 2015*, p. B52. Precession model of Capitaine, et al.
- [3] H.M. Nautical Almanac Office, Royal Greenwich Observatory and Nautical Almanac Office, U.S. Naval Observatory, *Planetary and Lunar Coordinates for the Years 1984-2000*, London and Washington, January 1983. See "Auxiliary Data," pp. 310-321. More accurate nutation models are available, but this one is deemed sufficient for the present application.
- [4] Urban, Sean, and P. K. Seidelmann, (Editors), *Explanatory Supplement to the Astronomical Almanac*, University Science Books, 3rd Edition (2013), p. 338.